
Zusammenfassung 

Seit SAS 9.1.3 sind die Anweisungen ODS DOCUMENT und die Prozedur DOCUMENT 
produktiv verfügbar. Aber vermutlich liegt es an den nicht direkt sichtbaren Ausgaben von 
ODS DOCUMENT, dass diese Anweisung nur selten eingesetzt wird. Dabei gestattet sie 
im Zusammenspiel mit der Prozedur DOCUMENT, die Ausgabeobjekte der Analysepro-
zeduren ohne deren erneute Ausführung in beliebiger Art und Weise anzuordnen, Teile 
auszuwählen und mit unterschiedlichen Stilen für unterschiedliche Ausgabeformate zu prä-
sentieren. 
In diesem Vortrag wird die Syntax der Anweisung ODS DOCUMENT aufgezeigt und die 
Möglichkeiten der Prozedur DOCUMENT an einfachen und auch komplexeren Beispielen 
dargestellt. 

 
Schlüsselwörter: ODS DOCUMENT, PROC DOCUMENT, REPLAY-Anweisung, 
HIDE-Anweisung 

1 Einleitung 
ODS DOCUMENT ist eine „Destination“, ein Kanal, der Anwendern das Leben er-
leichtern kann. Ausgaben von Prozeduren können damit reproduziert werden, ohne die 
Prozeduren erneut ausführen zu müssen. Die Ausgaben können dabei auch in anderer 
Reihenfolge sowie mit unterschiedlichen Stilen und Templates (Vorlagen) dargestellt 
werden. Weiterhin können Teile der ursprünglichen Ausgabe entfernt werden. 
Aufbauend auf einem „Dokument“ können somit verschiedene Reports für unterschied-
liche Benutzergruppen erzeugt werden. 
Mit diesem Vortrag werden die Grundlagen des ODS-Kanals eingeführt und der Ver-
such unternommen, das Interesse der Leser für weitere, eigene Experimente zu wecken. 
Die Idee für den Vortrag wurde [2], [3] und [4] entnommen, als Ergänzung diente [1]. 

1.1 Ein Beispiel 

Vielen SAS-Anwendern dürfte die folgende Vorgehensweise vertraut sein: Für eine 
SAS-Tabelle, hier: class aus der Bibliothek sashelp, sollen verschiedene Kennwerte 
(Häufigkeiten, Mittelwert und Korrelationen) berechnet werden. Dafür werden die drei 
Prozeduren FREQ, MEANS und CORR eingesetzt und folgendes Programm erstellt: 

  Proc Freq Data=sashelp.class;      * (1) ; 
     Tables sex; 
  Run; 
  Proc Means Data=sashelp.class; 
     Var age weight height; 
  Run; 
  Proc Corr Data=sashelp.class; 
     Var height*weight; 
  Run; 
 
Nachdem alles fehlerfrei läuft, müssen die Ergebnisse in Form eines PDF-Dokuments 
an den Vorgesetzten weitergeleitet. Dazu legt man eine ODS PDF-Klammer um die drei 
Prozedurschritte und führt folgendes Programm aus: 
 
ODS PDF File="meinpdf.pdf";          * (2) ; 
 
  Proc Freq Data=sashelp.class; 
     Tables sex; 
  Run; 
  Proc Means Data=sashelp.class; 
     Var age weight height; 
  Run; 
  Proc Corr Data=sashelp.class; 
     Var height*weight; 
  Run; 
 
ODS PDF Close; 
 
Nachdem der Vorgesetzte mit den Berechnungen einverstanden ist, wird das Ganze für 
die Veröffentlichung im RTF-Format benötigt. Also führt man das Programm nochmals 
aus, dieses Mal mit einer ODS RTF-Klammer: 
 
ODS RTF File="meinrtf.rtf";                  * (3) ; 
 
  Proc Freq Data=sashelp.class; 
     Tables sex; 
... 
  Proc Corr Data=sashelp.class; 
     Var height*weight; 
  Run; 
 
ODS RTF Close; 
 
Und schließlich, da der Stil nicht perfekt passt, eine erneute Ausführung mit einer 
STYLE-Option: 
 
ODS RTF File="meinrtf.rtf" Style=Journal;    * (4) ; 
... 
ODS RTF Close; 

Damit wurden die drei Prozeduren viermal ausgeführt (Programm 1-4), um die sichtba-
ren Ergebnisse in drei unterschiedlichen Weisen zu erhalten.  
 
Bei der verwendeten kleinen Tabelle sashelp.class geht das zügig. Bei großen Da-
tenmengen dagegen ist diese Art und Weise des Vorgehens sehr ineffizient. Und genau 
hier setzen ODS DOCUMENT und die Prozedur DOCUMENT an. 

1.2 Das erste Beispiel mit ODS DOCUMENT und PROC DOCUMENT 

 
Anstelle der ODS-Kanäle HTML, PDF oder RTF wird nun DOCUMENT verwendet: 
 
ODS DOCUMENT Name=work.meindoc;              * (5) ; 
 
  Proc Freq Data=sashelp.class; 
     Tables sex; 
  Proc Means Data=sashelp.class; 
     Var age weight height; 
  Run; 
  Proc Corr Data=sashelp.class; 
     Var height*weight; 
  Run; 
 
ODS DOCUMENT Close; 
 
Das Ergebnis ist damit kein HTML-, PDF- oder RTF-Dokument, sondern ein SAS-Ob-
jekt meindoc in der Bibliothek work. Um dieses anschließend darzustellen, nutzt man 
die Prozedur DOCUMENT mit der REPLAY-Anweisung (ohne weitere Argumente): 
 
  Proc Document Name=work.meindoc;      * (6) ; 
    Replay; 
  Run; 
  Quit; 
 
Die mittels ODS DOCUMENT erzeugten „Dokumente“ werden vom SAS System im 
binären Format in sogenannten Itemstores gespeichert. Diese sind  (unter Windows) an 
der Dateierweiterung sas7bitm erkennbar. 
 
Standardmäßig arbeitet ODS DOCUMENT im Update-Modus. D.h. wenn das Doku-
ment work.meindoc aus unserem Beispielprogramm 6 noch nicht existiert, wird dieses 
neu angelegt. Existiert es bereits, werden die Ergebnisse an die vorhandenen Prozedur-
ausgaben angehängt.  
Im Klartext bedeutet dies, dass wenn das Programm 5 ein weiteres, zweites Mal ausge-
führt wird, werden die Ergebnisse der Prozeduren FREQ, MEANS und CORR auch ein 
weiteres Mal abgelegt, insgesamt sind diese dann zweimal enthalten. 

Sollen die vorhandenen Ergebnisse stattdessen überschrieben werden, ergänzt man die 
Option WRITE in der PROC DOCUMENT-Anweisung:  
 
  Proc Document Name=work.meindoc(Write); 
 
Der Itemstore wird auf diesem Wege neu erstellt. 

2 Die Prozedur DOCUMENT 
Mit der Prozedur DOCUMENT und der Anwei-
sung REPLAY können die in einem Dokument 
enthaltenen Prozedurausgaben wiedergegeben 
werden. Um vorab aber einen Überblick über den 
Aufbau und den Inhalt des Dokuments (Item-
stores) zu erhalten, setzt man die Anweisung 
LIST ein: 
 
  Proc Document Name=work.meindoc; * 
(7); 
    List / Levels=All; 
  Run; 
  Quit; 
 
Damit erhält man eine tabellarische Aufstellung der in dem Dokument enthaltenen Ob-
jekte (Abb. 1). Die einzelnen Objekte werden dabei durchnummeriert (Spalte Obs) und 
jede Prozedur wird in einem eigenen Abschnitt (Directories/Ordner) aufgeführt (hier: 
\Freq#1, \Means#1 und \Corr#1). 
Die rechte Spalte Type gibt Auskunft, ob es sich um eine Tabelle oder um einen Ordner 
handelt. Man erkennt, dass \Freq#1 eine weitere Ordnerebene (Type=Dir) enthält: 
\Freq#1\Table1#1.  Erst darunter, also in Ebene 3 folgt die eigentliche Tabelle 
(Type=Table). 
Ohne die Angabe der Option LEVEL= würde nur die jeweils oberste Ebene angezeigt 
werden (gleichbedeutend mit LEVEL=1). Folglich entspricht in unserem Beispiel LE-
VEL=ALL genau LEVEL=3. 
 
In der mittleren Spalte von Abbildung 1 befinden sich die Pfadangaben (Path), die spä-
ter genutzt werden können, um einzelne Elemente oder Objekte auszuwählen. Die Nota-
tion der Elemente hat stets die Form: \directory#n1\directory#n2\object#n3, wobei die 
Ordner entweder für die Prozedur stehen (Zeile 1, 4, 6) oder für Abschnitte innerhalb 
der Prozedur (2) und die Objekte (3, 5, 7-9) analog ODS TRACE ON die Ausgabeob-
jekte beschreiben. 
 

Abbildung 1: Ergebnis der 
LIST-Anweisung 

Die Prozedur DOCUMENTS ist eine interaktive Prozedur, 
d.h. sie wird erst mit der Anweisung QUIT komplett been-
det (vgl. Programm 6). Mit der  Anweisung RUN können 
einzelne Anweisungen an das System zur Ausführung 
übergeben werden, ohne die Prozedur endgültig zu verlas-
sen (siehe unten in Abschnitt 2.6). 

2.1 Das Documents Fenster 

Eine interaktive Variante, um den Aufbau der Dokumente 
studieren zu können, bietet das Documents Fenster, das mit 
dem Befehl ODSDOCUMENTS geöffnet wird (Abb. 1). 
 
Neben dem Results und dem Explorer öffnet sich – sofern 
man die Standardeinstellungen für SAS unter Windows 
nutzt – das Documents Fenster als weitere Registerkarte. 
 
Die Ordner, Unterordner und Tabellen werden ähnlich dem 
Windows-Explorer dargestellt und können per Klick geöff-
net werden. 

2.2 Die REPLAY-Anweisung 

Mit der Anweisung REPLAY der Prozedur DOCUMENT 
können die Objekte eines Dokuments angezeigt werden. 
Ohne weitere Optionen werden alle im Dokument enthalte-
nen Objekte angezeigt, wie in Beispielprogramm 6. 
 
Enthält das Dokument allerdings zahlreiche Objekte, kann 
die Wiedergabe durch Benennung einzelner Objekte in der 
Replay-Anweisung beeinflusst werden. So liefert etwa 
 
     Replay \Corr#1; 
     Replay \Corr; 
 
nur die Objekte der Prozedur CORR. (Würden mehrere CORR-Ordner vorhanden sein, 
würde bei Aufruf ohne „#1“ (also zweite Replay-Anweisung) stets der erste Ordner an-
gezeigt werden.) 
 
     Replay \Corr#1\SimpleStats#1; 
 
liefert nur Tabelle der Kennwerte. Sollen mehrere Objekte dargestellt werden, trennt 
man die Objektnamen durch Kommata: 
 
     Replay \Corr#1, \Means#1; 

Abbildung 2: Das  
Documents Fenster 

2.3 Die HIDE-Anweisung 

Analog zu KEEP und DROP bei den Dateioptionen gibt es für die Prozedur 
DOCUMENT eine Anweisung, mit der einzelne Objekte von der Darstellung ausge-
schlossen werden können: HIDE. 
 
   PROC Document Name=work.meindoc; 
     Hide \Corr#1\VarInformation#1; 
   Run; 
 
Die Tabelle mit den Variableninformationen 
der Prozedur CORR wird hier ausgeschlossen. 
Gibt man die Objekte der Prozedur CORR nun 
aus mittels 
 
     Replay \Corr#1; 
 
erscheinen nur zwei Tabellen (nämlich Simp-
leStats und PearsonCorr) und im Log-Fenster 
folgende Meldung: 
 
NOTE: One or more user-hidden entries 
were excluded during document replay. 
 
Wiederholt man die Anzeige des Itemstore-Inhalts (mit der LIST-Anweisung), werden  
die „versteckten“ Tabellen mit einem Stern gekennzeichnet (Abb. 3). 
 
 
Wieder „sichtbar“ macht man solch versteckte Tabellen mit der Anweisung UNHIDE: 
 
     Unhide \Corr#1\VarInformation#1; 
 
Im Unterschied zu den Dateioptionen KEEP und DROP gilt HIDE über den Prozedur-
schritt hinaus, d.h. HIDE wirkt wie ein Schalter, der die Anzeige ein- oder ausschaltet. 

2.4 Ausgabe in mehrere Kanäle mit der Option DEST= 

Ein Dokument gleichzeitig in mehreren Ausgabeformaten abzulegen, eventuell mit un-
terschiedlichen Tabellen, gelingt mit der Option DEST= der Anweisung REPLAY. Zu-
nächst müssen dazu die Ausgabedateien definiert werden, hier PDF und RTF. 
 
ODS Pdf File="meinpdf.pdf"; 
ODS Rtf File="meinrtf.rtf"; 
 
  

Abbildung 3: LIST-Ausgabe mit 
„versteckten“ Objekten (HIDE) 

Danach folgt der Prozedur DOCUMENT-Schritt: 
 
PROC Document Name=work.meindoc; 
 
  Replay \Means#1 / Dest=(Rtf); 
  Replay \Corr#1  / Dest=(Rtf Pdf); 
  Replay \Freq#1  / Dest=(Pdf); 
 
Run;Quit; 
 
Hierbei werden die MEANS- und CORR-Tabellen in das RTF-Dokument überführt, die 
CORR- und FREQ-Tabellen in das PDF-Dokument. 
 
Und als letztes werden die ODS-Klammern geschlossen: 
 
ODS Pdf Close; 
ODS Rtf Close; 
 
Analog funktioniert das Vorgehen auch, wenn zwei RTF-Dokumente erzeugt werden 
sollen, etwa mit unterschiedlichen Stilen wie „barretsblue“ und „minimal“. Man ver-
wendet dabei die Option ID= der ODS RTF-Anweisung, um die einzelnen RTF-Doku-
mente zu bezeichnen: 
 
ODS RTF (Id=1) File="test1.rtf" Style=Barrettsblue; 
ODS RTF (Id=2) File="test2.rtf" Style=Minimal; 
 
   Proc Document Name=work.meindoc; 
     Replay        / Dest=(Rtf (Id=1)); 
     Replay \means / Dest=(Rtf (Id=2)); 
   Run;Quit; 
 
ODS RTF (Id=1) Close; 
ODS RTF (Id=2) Close; 

2.5 Objekte auswählen mit Hilfe von WHERE 

Analog zur Dateioption WHERE= können auch bei der Wiedergabe mit REPLAY die 
Objekte nach Bedingungen ausgewählt werden. Dazu können BY-Gruppenbedingungen 
verwendet werden oder Objekteigenschaften, die im ODS-Dokument hinterlegt sind. 
 
Die Anweisung DETAILS der LIST-Anweisung zeigt einige, bei weitem nicht alle, 
Objekteigenschaften an (Abb. 4): 
 
   PROC Document Name=co.meindoc; *   (7); 
     List / Details Levels=All; 
   Run;Quit; 

Abbildung 4: Detailansicht der Elemente im Itemstore 

Bei den Eigenschaften handelt es sich um:  
_name_, _path_, _label_, _labelpath_, _type_ sowie   
_min_, _max_, _obs_, _cdate_, _mdate_, _ctime_, _mtime_ 
 
Damit können dann beispielsweise folgende Bedingungen formuliert warden: 
 
   PROC Document Name=work.meindoc;                  * (8) ; 
     Replay Freq(Where=( name_="OneWayFreqs")); 
     Replay \ (Where=( name_="OneWayFreqs")); 
     Replay \ (Where=( labelpath_ ? "Sex"));  
     Replay Means#2(Where=( sex="M")); 
   Run; 
   Quit; 
 
Anmerkung: Um die vierte REPLAY-Anweisung in Programm 8 fehlerfrei ausführen zu 
können, muss der PROC MEANS-Schritt aus Programm 5 erweitert werden um die 
Anweisung „Class sex;“ und erneut ausgeführt werden: 
 
ODS DOCUMENT Name=work.meindoc;              * (9) ; 
  Proc Means Data=sashelp.class; 
     Var age weight height; 

     Class sex; 
  Run; 
ODS DOCUMENT Close; 
 
Mit den bislang gezeigten Anweisungen und Optionen wurden noch nicht alle Möglich-
keiten, welche die Prozedur DOCUMENT bietet vorgestellt. Eine letzte, die in diesem 
Beitrag noch präsentiert werden soll, erinnert uns an alte DOS-Zeiten. 

2.6 Eigene Reports innerhalb des Dokuments erstellen 

Mit Hilfe der REPLAY-Anweisung haben wir in den bisherigen Beispielen die Proze-
durausgaben wiedergegeben, mit HIDE einzelne Tabellen ausgeblendet. Wir können 
aber auch eigene Reports innerhalb eines Doku-
ments erstellen und darin einzelne Tabellen aus 
dem Dokument einfügen. Wie das gehen kann, 
soll das folgende Beispiel illustrieren: 
 
PROC Document Name=work.meindoc; 
  Make meinreport; 
  Run; 
 
Mit der MAKE-Anweisung wird ein neuer Ord-
ner in dem Dokument meindoc angelegt, hier 
unter dem Namen meinreport, vergleichbar mit 
dem DOS-Kommando make oder dem Unix-Be-
fehl md. 
 
Tipp: Wenn Sie die Anweisungen ausprobieren 
möchten, führen Sie immer mal wieder ein List 
/ Levels=All; aus, um zu sehen, was sich im 
Dokument verändert hat (Abb. 4). 
 
  Dir \Freq#1; 
  Run; 
 
Die Anweisung DIR erlaubt die Ordner zu 
wechseln, analog dem DOS dir-Kommando oder 
dem Unix-Befehl cd. In obigem Beispiel wird in 
den Ordner \Freq#1 gewechselt (Abb. 6). 
 
  Copy Table1#1 to \meinreport#1; 
  Run; 
Quit; 
 
Im letzten Schritt wird nun die Tabelle mit den 
Häufigkeiten in den Ordner meinreport kopiert. 

Abbildung 5: Itemstore mit 
eigenem Ordner meinreport 

Abbildung 6: Eingeschränkte 
Sicht (nur Ordner \Freq#1) 

  Dir \; 
  List / Levels=All; 
  Run; 
Quit; 
 
Die Anweisung Dir \ springt wieder in die 
oberste Ebene des Dokuments, so dass das in 
Abb. 7 gezeigte Ergebnis erscheint. 
 
Neben den vorgestellten Anweisungen gibt es 
noch zahlreiche weitere, wie RENAME zum 
Umbenennen der Ordner und Elemente, 
MOVE zum Verschieben von Objekten, DE-
LETE zum Löschen und LINK um symboli-
sche Links zu definieren. Details können 
bspw. In [1] nachgelesen werden. 
 
Zum Schluss soll noch ein Beispiel gezeigt werden, wie die kopierten Tabellen (zusätz-
lich zu den Häufigkeiten wurde auch noch die Variableninformation der Prozedur 
CORR in meinreport übertragen) mithilfe der Anweisungen OBTITLE, OBSTITLE und 
OBPAGE verändert werden können, ohne die Prozeduren FREQ und CORR selbst 
nochmals ausführen zu müssen. 

  Obtitle \meinreport#1\Table1#1\OneWayFreqs#1 ; 
  Obtitle \meinreport#1\VarInformation#1 ; 

OBTITLE entfernt den Titel, analog zu TITLE. 

  Obstitle \meinreport#1\Table1#1\OneWayFreqs#1  
     "Mein eigener Report mit den Häufigkeiten"; 
  Obstitle \meinreport#1\VarInformation#1  
     "Infos zu den Variablen"; 

Mit der Anweisung OBSTITLE definiert man neue Untertitel. 

  Obpage \meinreport#1\VarInformation#1 / Delete; 
 
Die Anweisung OBPAGE entfernt 
Seitenumbrüche, die standardmäßig 
zwischen zwei Prozeduren vom Sys-
tem eingefügt werden. (In Abb. 4, 
rechte Spalte wurden diese angezeigt.) 

Beide Tabellen erscheinen damit di-
rekt untereinander mit den selbst de-
finierten Überschriften (Abb. 8). 

 

Abbildung 7: Kompletter Inhalt 
des Itemstores 

Abbildung 8: Eigener Report mit 
Überschriften und ohne Seitenumbruch 

3 Fazit 
Die ODS DOCUMENT Anweisung gestattet im Zusammenspiel mit der Prozedur 
DOCUMENT eine perfekte Trennung zwischen der Durchführung einer Analyse oder 
Berechnung und der Darstellung der Ergebnisse. Wiederholtes Ausführen von Proze-
durschritten zur Optimierung der Präsentation ist nicht mehr notwendig. Programme 
und Arbeitsabläufe können dadurch effizienter gestaltet werden. Ebenso können Grafi-
ken verarbeitet werden. 

4 Ausblick 
Mit dieser Einführung in die Funktionalitäten sollte die erste Hürde genommen und die 
Lust geweckt worden sein, selbst eigene Erfahrungen mit der Anweisung ODS 
DOCUMENT und der Prozedur DOCUMENT zu sammeln. So würde die Autorin inte-
ressieren, wie erfahrene Makroprogrammierer mit der Situation umgehen, dass die Ob-
jekte in den Dokumenten lediglich durchnummeriert werden. Hier wäre eine Methode 
von Interesse, wie die Objekte so bezeichnet werden können, dass sie programmtech-
nisch wieder auffindbar sind. 

Vielleicht bietet sich auf der nächsten KSFE Gelegenheit, die Erfahrungen im Rahmen 
einer Diskussionsrunde oder als Tipp und Trick auszutauschen. 

